

Next: Missing values Up: Working with time-series data Previous: Avoiding the lag() function

Rolling regressions without macros

<u>Boehmer, Broussard, and Kallunki (2002)</u> recommend using macros to run rolling regressions. While macros make impossible tasks possible, they aren't particularly efficient. I describe here a macro-independent way of running rolling regressions, and doing similar tasks.

The task is to run rolling 24-month regressions of monthly stock excess returns on the Fama-French-Carhart factors. That is, at the end of each month, I need to:

- Get the last 24 months of excess returns for each stock, checking that there are at least, say, 12 months of data available.
- Get the Fama-French-Carhart factors.
- Run rolling regressions

I first create a dataset containing "rankdates", which are the date identifiers for the rolling regression. A rankdate of 31Dec2001, for instance, uses data from 31Jan2000 to 31Dec2001, inclusive.

To do this, I first get the first and last date for each permno, and then create a complete list by filling in the in-between dates. I do this because if I were to just use the permno-date pairs available in MSF, if a return is missing for a particular month, then that month will not be a rankdate, even though it might be perfectly valid to make that date a rankdate.

```
data firstandlastdates;
   set crsp.msf(keep=permno date);
   by permno; /*MSF is always sorted by permno date*/
   retain firstdate;
   date=intnx('month', date, 1)-1;
   if first.permno then firstdate=date;
   if last.permno then do;
       lastdate=date;
       output;
   end;
data permnosrankdates(rename=(date=rankdate));
    set firstandlastdates;
    date=firstdate;
    do while(date<=lastdate);</pre>
       output;
       date=intnx('month', date+1, 1)-1;
    end:
run;
```

For each rankdate, I then get the list of the 24 dates from which that rankdate will use data.

```
data permnosrankdates;
  set permnosrankdates;
  date=rankdate;
  i=1;
  do while(i<=24);
     output;</pre>
```

```
\begin{array}{c} \text{date=intnx('month', date, 0)-1;} \\ \text{i=i+1;} \\ \text{end;} \\ \text{run;} \end{array}
```

permnosrankdates is a dataset that looks, in part, like this:

0bs	PERMNO	rankdate	date	i
0bs 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19	PERMNO 10000 10000 10000 10000 10000 10000 10000 10000 10000 10000 10000 10000 10000 10000 10000 10000 10000	rankdate 19851231 19851231 19851231 19851231 19851231 19851231 19851231 19851231 19851231 19851231 19851231 19851231 19851231 19851231 19851231 19851231 19851231 19851231 19851231	date 31DEC1985 30N0V1985 310CT1985 310CT1985 31SEP1985 31JUL1985 30JUN1985 31MAY1985 31MAR1985 31MAR1985 31JAN1985 31DEC1984 30N0V1984 310CT1984 30SEP1984 31JUL1984 31JUL1984 30JUN1984	i 1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17 18 19
20	10000	19851231	31MAY1984	20
21		19851231	30APR1984	21
22	10000	19851231	31MAR1984	22
23	10000	19851231	29FEB1984	23
24	10000	19851231	31JAN1984	24
25	10000	19860131	31JAN1986	1
26	10000	19860131	31DEC1985	2
27	10000	19860131	30N0V1985	3
28	10000	19860131	310CT1985	4
29	10000	19860131	30SEP1985	5
30	10000	19860131	31AUG1985	6

We don't need to keep i, but I kept it for clarity.

Once we have this, all we need to do is merge it with the factors and the returns:

```
data ff;
    set ff.factors_monthly(keep=date rf smb hml umd mktrf);
    date=intnx('month', date, 1)-1;
proc sort data=permnosrankdates;
    by date permno;
data permnosrankdates;
    merge permnosrankdates(in=a) ff(in=b);
    by date;
    if a and b;
run;
data msf;
   set crsp.msf(keep=permno date ret);
   where ret is not missing;
   date=intnx('month', date, 1)-1;
proc sort data=msf;
    by date permno;
run;
```

```
/*permnosrankdates is already sorted*/
data permnosrankdates;
   merge permnosrankdates(in=a) msf(in=b);
   by date permno;
   if a and b;
run;
```

Notice that I merged by date, not rankdate.

And now all that remains is to calculate excess returns and run the regressions:

```
data permnosrankdates;
    set permnosrankdates;
    exret=ret-rf;
run;

proc sort data=permnosrankdates;
    by permno rankdate;

proc reg data=permnosrankdates outest=est edf;
    by permno rankdate;
    model exret=mktrf smb hml umd;
run;
```

Notice I run proc reg by rankdate, not date.

Running proc print on EST yields the following output:

0bs	PERMNO	rankdate	_MODEL_	_TYPE_	_ DEPVAR_	RMSE	_ Intercept	mktrf
1 2 3 4 5 6 7 8 9 10 11 12 13 14 15 16 17	10000 10000 10000 10000 10000 10000 10000 10000 10000 10000 10000 10000 10000 10000	19860228 19860331 19860430 19860531 19860630 19860731 19860930 19861031 19861130 19861231 19870131 19870228 19870331 19870430 19870531 19870630	MODEL 1 MODEL 1	PARMS	exret exret exret exret exret exret exret exret exret exret exret exret exret exret exret exret exret exret	0.4871 0.3447 0.2911 0.2521 0.2580 0.2726 0.2525 0.2382 0.2421 0.2297 0.2205	6 -0.1872 9 -0.1559 7 -0.1216 6 -0.1214 9 -0.0767 6 -0.1228 1 -0.1217 8 -0.1090 4 -0.1254 2 -0.1194	-32.219 -31.110 2.102 -191.074 -4.129 -2.784 -0.492 -0.499 -1.790 -0.709 -0.665 -0.784 -1.872 -1.942 -1.983
0bs	smb	hı	ml	umd	exret _	_IN	PEDF_	_RSQ_
1 2 3 4 5 6 7 8 9 10 11 12 13 14 15	0.000 0.000 -71.326 340.707 -884.257 -13.672 -8.529 7.327 7.303 3.312 4.391 4.378 4.342 0.837 0.669	0.0 0.1 477.8 -1322.8 -16.5 -10. 3.3 -0.8 0.9 -0.9 -3.5 -3.5	00 0 000 0 88 0 51 506 38 7 71 5 24 4 22 4 84 3 97 3 91 3 04 3 20 2	.000 .000 .000 .000 .315 .106 .470 .754 .751 .657 .748 .725 .353 .548	-1 -1 -1 -1 -1 -1 -1 -1 -1 -1	2 3 4 4 4 4 4 4 4 4 4 4 4	1 0 2 0 3 0 4 0 5 0 5 1 5 2 5 3 5 4 5 5 5 6 7 8 5 9	1.00000 1.00000 1.00000 1.00000 0.05020 0.54571 0.51848 0.52996 0.42145 0.30088 0.30569 0.31433 0.26778 0.27495

16	0.604	-3.41	2.454	-1	4	5	11	0.27148
17	0.604	-3.41	2.454	-1	4	5	11	0.27148

We can tell how many observations we have for each date by looking at the _EDF_: the error degrees of freedom. We asked for this by including edf in the proc model statement. If we want to keep only regressions for which there were 12 observations or more, we keep those where _EDF_>=7.

EST is the rolling regression output dataset.

Andre de Souza 2012-11-19

This technique has wide application: rolling regressions, obviously, but also rolling standard deviations, moving averages and whatnot.

Next:	Missing v	<u>ralues</u> U	o: <u>Working</u>	with	time-series	data	Previous:	Avoiding	the lag()	<u>function</u>